

```
prefix rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
prefix ao: <http://www.dei.unipd.it/database2/airbnbOntology#>
prefix rdfs: <http://www.w3.org/2000/01/rdf-schema#>
prefix xsd: <http://www.w3.org/2001/XMLSchema#>
```

#1: Which listing type is the most repeated one in all boroughs?

```
select ?borough_name ?listing_type_name (COUNT (?Listing) as ?listing_count) where
{
    ?Listing ao:hasType ?listingType.
    ?listingType ao:listingTypeName ?listing_type_name.
    ?Listing ao:locatedIn ?Area.
    ?Area rdfs:subClassOf ?Borough.
    ?Borough ao:boroughName ?borough_name.
    {
        select ?Borough (MAX(?count) as ?max_count) where
        {
            select ?Borough ?listingType (COUNT (*) as ?count) where
            {
                ?Listing ao:hasType ?listingType.
                ?Listing ao:locatedIn ?Area.
                ?Area rdfs:subClassOf ?Borough.
            }
            GROUP BY ?Borough ?listingType
        }
        GROUP BY ?Borough
    }
}
GROUP BY ?borough_name ?listing_type_name ?max_count
HAVING (?listing_count = ?max_count)
```

#2: Do entire home/apt have more availability than private room in a year?

ASK where

```
{
  {
    select ?listingType (SUM(?availability_per_listing) as
?sum_available_days_private) where
    {
      select ?Listing ?listingType ?availability_per_listing where
      {
        ?Listing ao:hasType ?listingType.
        ?listingType ao:listingTypeName "Private room"^^xsd:string.
        ?Listing ao:availability365 ?availability.
        BIND ("365"^^xsd:int - ?availability AS ?availability_per_listing)
      }
    }
    GROUP BY ?listingType
  }
  {
    select ?listingType (SUM(?availability_per_listing) as
?sum_available_days_entire) where
    {
      select ?Listing ?listingType ?availability_per_listing where
      {
        ?Listing ao:hasType ?listingType.
        ?listingType ao:listingTypeName "Entire home/apt"^^xsd:string.
        ?Listing ao:availability365 ?availability.
        BIND ("365"^^xsd:int - ?availability AS ?availability_per_listing)
      }
    }
    GROUP BY ?listingType
  }
  FILTER (?sum_available_days_entire > ?sum_available_days_private)
}
```

#3: The total number of listings not located in Brooklyn

```
select (count(?Listing) as ?listing_count) where
{
  ?Listing rdf:type ao:Listing.
  MINUS
  {
    ?Listing ao:locatedIn ?Area.
    ?Area rdfs:subClassOf ?Borough.
    ?Borough ao:boroughName "Brooklyn".
  }
}
```

#4: Mean review score for each area

```
select ?borough_name (count(?review_rating) as ?count) (sum(?review_rating)/?count as
?review_mean) where
{
  ?Listing ao:reviewRating ?review_rating.
  ?Listing ao:locatedIn ?Area.
  ?Area rdfs:subClassOf ?Borough.
  ?Borough ao:boroughName ?borough_name.
}
GROUP BY ?Borough ?borough_name
ORDER BY DESC (?review_mean)
```

#5: Host has listing in an area

```
CONSTRUCT { ?Host ao:hasListingLocatedIn ?Area }
WHERE
{
  ?Listing ao:belongsTo ?Host;
    ao:locatedIn ?Area.
  ?Area rdfs:subClassOf ?Borough.
  ?Borough ao:boroughName ?borough_name.
}
```

#6: Find the hosts with most reviews and the one with least

```
select ?Host ?review_count where
{
  {
    select ?Host (count(?Review) as ?review_count) where
    {
      ?Review ao:referredTo ?Listing.
      ?Listing ao:belongsTo ?Host.
    }
    GROUP BY ?Listing ?Host
    ORDER BY DESC(?review_count)
    LIMIT 1
  }
  UNION
  {
    select ?Host (count(?Review) as ?review_count) where
    {
      ?Review ao:referredTo ?Listing.
      ?Listing ao:belongsTo ?Host.
    }
    GROUP BY ?Listing ?Host
    ORDER BY ASC(?review_count)
    LIMIT 1
  }
}
```

#7: Among all listings of the host having maximum listings, which listing has got more reviews?

```
select ?Listing ?Host (count(?Review) as ?review_count) where
{
    ?Review ao:referredTo ?Listing.
    ?Listing ao:belongsTo ?Host.

    FILTER (?Host = ?inner_host) # host107434423
    {
        select ?inner_host (count(?Listing) as ?listing_count) where
        {
            ?Listing ao:belongsTo ?inner_host.
        }
        GROUP BY ?inner_host
        ORDER BY DESC(?listing_count)
        LIMIT 1
    }
}
GROUP BY ?Listing ?Host
ORDER BY DESC(?review_count)
LIMIT 1
```

#8: Find the host with the most listings for each borough

```
select ?borough_name ?Host (count(?Listing) as ?listing_count) where
{
    ?Listing ao:belongsTo ?Host.
    ?Listing ao:locatedIn ?Area.
    ?Area rdfs:subClassOf ?Borough.
    ?Borough ao:boroughName ?borough_name.
    {
        select ?Borough (max(?count) as ?max_count) where
        {
            select ?Host ?Borough (count(?Listing) as ?count) where
            {
                ?Listing ao:belongsTo ?Host.
                ?Listing ao:locatedIn ?Area.
                ?Area rdfs:subClassOf ?Borough.
            }
            GROUP BY ?Host ?Borough
        }
        GROUP BY ?Borough
    }
}
GROUP BY ?borough_name ?Host ?max_count
HAVING (?listing_count = ?max_count)
```

#9: What is the percentage of each listing type in a borough?

```
select ?listingType ?inner_borough_name (count(?Listing)/?total as ?percentage) where
{
  ?Listing ao:hasType ?listingType.
  ?Listing ao:locatedIn ?Area.
  ?Area rdfs:subClassOf ?Borough.
  ?Borough ao:boroughName ?borough_name.
  {
    select ?inner_borough_name (count(?Listing) as ?total) where
    {
      ?Listing rdf:type ao:Listing.
      ?Listing ao:locatedIn ?Area.
      ?Area rdfs:subClassOf ?Borough.
      ?Borough ao:boroughName ?inner_borough_name.
    }
    GROUP BY ?inner_borough_name
  }
}
GROUP BY ?listingType ?borough_name ?total ?inner_borough_name
HAVING (?borough_name = ?inner_borough_name)
ORDER BY ?listingType DESC(?percentage)
```

#10: Which area in Manhattan has the maximum average price of private room?

```
select ?area_name (AVG (?price) as ?max_average_price) where
{
    ?Listing ao:price ?price.

    ?Listing ao:hasType ?listingType.
    ?listingType ao:listingTypeName "Private room"^^xsd:string.

    ?Listing ao:locatedIn ?Area.
    ?Area ao:areaName ?area_name.
    ?Area rdfs:subClassOf ?Borough.
    ?Borough ao:boroughName "Manhattan"^^xsd:string.
}
GROUP BY (?area_name)
ORDER BY DESC(?max_average_price)
LIMIT 1
```